
When you plan your upgrade to
z/OS, consider upgrading the
in-house assembler routines

you maintain to take advantage of new
instructions and assembler facilities. It is
often difficult to embrace change when it
requires you to dust off your assembler skills
or crack the IBM Principles of Operations
book to learn how to use new instructions.
However, you could consider a simple first
step in an upgrade path for these programs.
Modernize only the housekeeping “stubs”
that you use to establish and restore your
program’s execution environment. In other
words, take a phased approach to altering
your routines and reduce the exposure to
error that change introduces. Are there
reasons for considering modification at all?

Yes. In fact, if you review the z/OS version
of the IBM Principles of Operations you will
discover recommendations regarding the
reasons to replace several older instructions
that your routines have been using for years.
Bear in mind that the shareware programs
you may be using may not have kept pace
with operating system changes and enhance-
ments. This means you have an opportunity
to update your routines and to implement the
new instruction sets that have come with the
evolution of MVS, the G5 series machines
and z/OS. Do not fret. You can meet the
challenge to change these routines and prepare
them for z/OS even if your skill set is not
what it used to be.

This article will present, in an evolutionary
fashion, several programming housekeeping
samples that highlight the progress of the
architectural changes from MVS to z/OS. In
addition to providing some background

information on the instructions used, I will
also supply in Part II several code samples
that use new features of the High Level
Assembler (HLA) that include the use of the
HLA ToolKit structured macros.

PREPARING YOUR PROGRAMS
FOR Z/OS

First, a brief history is in order. ESA/370
(Enterprise Systems Architecture) introduced

a new execution environment (architecture)
and a new set of instructions and registers
in its support. Access Register (AR) mode
is the new execution mode of operation that
provides an environment for programs to
access address spaces provided by the OS
architecture. While in AR mode, programs
can perform system-level functions such as
mapping the storage of other address
spaces, defining and manipulating data
spaces, and other OS features that have

Assembler Coding in the
z/OS Environment: Part I

This article presents several z/OS assembler programming housekeeping samples and examines
assembler instructions that highlight the progress of the architectural changes since MVS/370.

TECHNICAL SUPPORT • FEBRUARY 2002

ROUTINE AMODE 31 Program addressing mode
ROUTINE RMODE 24           Program execution locations
ROUTINE CSECT               Establish addressibility

BAKR  R14,0          Store Regs and environment on linkage-stack
LR    R12,r15           R12 will be the base reg
USING Routine,R12       Addressability to the routine
LA    R13,MySave_Area    Load the addr of my save area
MVC   04(04,R13),=CL04’F1SA’  Note to the OS that the linkage-stack is in

use.
*  Your routine

Pr     ,                “Pop” the linkage-stack
MySave_area    DS  18F  � Save area

LTORG ,               Drop the literal Pool
END   ROUTINE          Routine termination

� IBM recommends that you supply a minimum 18 full-word save area pointed to by GPR R13 to remain compatible
with IBM’s Language Environment (LE).

FIGURE 1: A BAKR/PR PRODUCTION “STUB”

ROUTINE AMODE 31
ROUTINE RMODE 24
ROUTINE CSECT

BAKR  R14,0
LAE   R12,00(R15,00) Prepare base register for an ARMODE caller

USING Routine,R12
LA    R13,MySave_Area
MVC   04(04,R13),=CL04’F1SA’

*  Your routine
Pr     ,

MySave_area    DS  18F
LTORG ,
END   ROUTINE

FIGURE 2: BAKR/PR PROGRAMMING STUB WITH USE OF LAE

evolved over the years. An AR execution
environment is usually not necessary for
problem state programs, but several of those
new instructions used to support that envi-
ronment present a compelling reason to
implement them. In the next several sections,
I will discuss how to use these instructions
in the context of assembler programming
housekeeping activities.

BAKR/PR INSTRUCTION
IMPLEMENTATION

ESA/370 introduced the Branch and
Stack (BAKR) and Program Return (PR)
instruction pair. Respectively, they save and
recall all General Purpose Registers (GPR)
and AR (the new registers in ESA/370), the
Program Status Word (PSW), and store pro-
gram status and other environmentally related
values on a “push” type linkage-stack. The
use of BAKR does not change your program’s
addressing execution mode or alter the logic
in any way. BAKR operates similarly to
BASR or BALR in housekeeping type
operations and remembers the addressing
mode of the caller (PSW Bit 32). Likewise,
the PR instruction restores the caller’s
execution state, key, GPRs and ARs. The
BAKR/PR combination is only available to
primary-space and access-register programs
— a mode in which the majority of most
programs are written.

The value of using the BAKR/PR combi-
nation obviously surpasses the functionality
provided by IBM’s housekeeping macros
SAVE and RETURN. With BAKR/PR, your
programs do not have to remember the
sometimes confusing sequence of register
conventions such as the “backward/for-
ward” chaining sequences for register save
areas. Moreover, there are benefits to using
these two instructions in improved program
environment maintenance, program read-
ability, and preparation for future OS
release compatibility. Take advantage of
these two instructions now!

Some commentators have suggested that
the BAKR/PR combination introduces
excessive overhead as compared to using
traditional SAVE and RETURN. I do not
plan to debate the issue here. I suggest you
use the BAKR/PR combination only at
program start and termination and not as a
convention for use in subroutines. I have
other ideas on subroutine housekeeping that
I will explain in Part II. Inconsistent or
unmatched BAKR/PR use in subroutines

can have dire consequences such as system
ABENDs. If you choose to use BAKR/PR
in a subroutine, be sure not to branch out of
the routine. Figure 1 presents a brief example
of the use of the BAKR/PR combination in
a programming stub.

THE USE OF LAE

You should consider using other ESA/370
instructions when updating your housekeep-
ing routines to support z/OS. One instruction
is the implementation of Load Address
Extended (LAE). LAE differs from the Load
Address (LA) as it not only loads an address
in a specified target register in the same way
that LA does, but it also operates on the cor-
responding AR. The LAE informs the corre-
sponding AR how to resolve addressing of
local storage addresses. Therefore, when
establishing a base register for your program,
you may want to implement the LAE
instruction in case the caller of your routine
is operating in AR mode or your routine uses
AR mode services. Figure 2 provides an
example that was built upon Figure 1.

THE USE OF EREG,
MSTA AND ESTA

The extracted stacked registers (EREG)
instruction enables a program to extract a
register pair from the most recently estab-
lished linkage stack after it issues a BAKR
instruction. If you use BAKR only once at

the start of the program, it is simple to
extract the contents of R1 that may contain
a passed parameter any time during pro-
gram execution. Therefore, if the caller of
your routine passes a parameter list or a
storage address in R1 for your use, the
instruction “EREG R0,R1” will extract
GPR and AR R0 and R1 values from the
most current linkage stack and overlay the
current values.

Also, if you want to be clever and for-
ward thinking, consider using another
instruction available to you: the linkage
stack instructions Modify Stack State
(MSTA) and Extract State (ESTA). When
BAKR establishes a linkage stack, an 8-byte
user modifiable area is made available for
you to use as you see fit. You can store any
information you want in this area: perhaps
your girlfriend’s phone number in binary?
Alternatively, consider this: When a PR
instruction is executed, it restores the caller’s
environment including registers 2 - 14 but
not R0, R1, R14, and R15. What can you
do if you wanted to return the caller’s
original value stored in R1? You can use
MSTA and ESTA in combination to
accomplish this.

Issue an MSTA instruction before R1 is
altered to store the contents of R0 and R1 in
the modify stack state area. Then use ESTA
to extract the original R0 R1 combination at
the end of your routine before you return to
the calling routine or OS. Be advised that
MSTA does use a GPR even-odd register

ROUTINE AMODE 31
ROUTINE RMODE 24
ROUTINE CSECT

BAKR R14,0
MSTA R0       Store R0 and R1 in modification area
LAE  R12,00(R15,00)
USING Routine,R12
LA   R13,MySave_Area
MVC 04(04,R13),=CL04’F1SA’

*  Your routine
LA   R0,3       Say what we want to get back from the linkage stack
ESTA R0,R0     Restore R0 and R1 from the MSTA
pr     ,      Return to caller or OS

MySave_area   DS  18F
LTORG ,
END ROUTINE

FIGURE 3: BAKR/PR PROGRAMMING STUB WITH USE OF MSTA

BAL   OPSYN BAS         * Change all occurrences of BAL to BAS
BALR OPSYN BASR        * Change all occurrences of BALR to BASR

FIGURE 5: OPSYN USAGE 

LM  R0,R1,=CL08’ROUTINE’
MSTA R0               Store R0 and R1 in modification area

FIGURE 4: MSTA NOTATION 

pair. Figure 3 provides a simple way to
accomplish this without using any program
storage: store R1 and then recall it at the end
of a program.

You could also improve your debugging
of a routine by placing an 8-byte character
string in the MSTA area. For example, the
eight bytes could be your subroutine name
or program name. When evaluating an
ABEND, you can see what was in the
MSTA area to determine which subroutine
was active at the time. See Figure 4.

REPLACING BAL/BALR WITH
BAS/BASR

As you continue updating your programs
for z/OS compatibility, consider exchanging
the use of BAL and BALR for Branch and
Save (BAS) and Branch and Save Register
(BASR). Why should you consider this
exchange? The IBM Principles of Operations
for z/OS states the following:

“The BRANCH AND LINK instruction
(BAL and BALR) is provided for compatibility
purposes. It is recommended that, where
possible, the BRANCH AND SAVE instruc-
tion (BAS and BASR), BRANCH RELATIVE
AND SAVE, or BRANCH RELATIVE AND
SAVE LONG be used and BRANCH AND
LINK be avoided, since the latter places
non-zero information in bit positions 32-39
of the link register in the 24-bit addressing
mode, which may lead to problems.
Additionally, in the 24-bit addressing mode,
BRANCH AND LINK may be slower than
the other instructions because BRANCH
AND LINK must construct the ILC, condi-
tion code, and program mask to be placed in
bit positions 32-39 of the link register.”

Only the instruction needs to change and
not the operands when converting from
BAL/BALR to the BAS/BASR format.
Remember that upon returning from a sub-
routine to your program, use the “BSM
0,Rx” where Rx is the register you used to
branch to your subroutine.

You can implement a temporary change
in your program by using the OPSYN
assembler instruction. Since only the BALR
and BAL instructions are changed, the
OPSYN provides a simple way to imple-
ment this update. Use the two lines in
Figure 5 as the first two lines of your pro-
gram, assemble and then test. This will
work as long as you do not perform any
conditional branching upon the immediate
return from a BAL or BALR. If you use

OPSYN in your routines, place the return
code of your subroutine in R15, then upon
return from a BAL or BALR perform a “LTR
R15, R15”, and then add the branch logic you
have to the back end of the LTR instruction.

Let’s construct a simple example to
demonstrate what I have discussed so far. A
subroutine is called using BAS where R14
is used as the branch register. Figure 6 pre-
sents a sample program “stub” that includes
the use of BAKR/PR, LAE, MSTA, and
ESTA instruction.

PROGRAMMING “STUB”
EVOLUTION

I have covered the majority of new
instructions to use when you are building a
new programming “stub” for your z/OS
conversion. In Part II of this article, I will
bring together the various elements dis-
cussed so far and explore methods used in
assembler housekeeping since 370/MVS.
Each example presents a progression, in an
evolutionary fashion, of the impact that the
hardware architecture has had on assembler
housekeeping. The progressive examples
will help you to recognize programs that
need updating and hopefully give you a bet-
ter understanding of the reason why pro-
grams were designed they way they were.

MVS “STUB”

The example in Figure 7 presents a pro-
gram stub that was written before

MVS/XA but will still work today assuming
AMODE(31) execution is not required.
Subsequent examples evolve to use the
new instructions and the HLA to operate
in the ESA/370 and z/OS environments.
Several conventions are employed such
as the use of GPR 12 as the base address-
ing register for the stub, although this is
not a requirement.

MVS/XA “STUB”

For completeness, let’s examine how
the same program stub, used in the exam-
ple in Figure 8, could be coded using the
370/XA instruction set and environment.
This iteration of the programming stub
provides for AMODE(31) execution and
returns the caller in the correct addressing
mode via the BSM instruction. Simply
specifying “AMODE 31” at the start of
the program does not ensure it will exe-
cute in AMODE(31) especially if another
routine in a different addressing mode
called your routine. This routine uses a
convention to load a 31-bit address into
register 14 and then issues the BSM
instruction to ensure 31-bit addressing.
After the BSM, you will replace R14
(31-bit version) in the savearea.

The IBM RETURN macro is also
replaced with code that matches the
RETURN macro except in the last
instruction where a “BSM 0,R14” is used
instead of “BR R14”. The Branch and Set
Mode (BSM) returns to the caller in the

BAL  OPSYN BAS
BALR  OPSYN BASR
ROUTINE AMODE 31
ROUTINE  RMODE 24
ROUTINE  CSECT

BAKR R14,0
MSTA R0           Store R0 and R1 in modification area
LAE  R12,00(R15,00)
USING Routine,R12
LA  R13,MySave_Area
MVC 04(04,R13),=CL04’F1SA’
BAS R14,SubRoutine Go do something

*  Your routine
LA  R0,3           Say what we want
ESTA R0,R0          Restore R0 and R1
pr  ,
DS 0F

SubRoutine  EQU *
STM  R14,r2,SubRoutine_save

*  Do something
LM  R14,r2,SubRoutine_save
BSM  0,R14      Return to caller in callers mode

SubRoutine_save DS 10F
MySave_Area   DS  18F

LTORG ,
END ROUTINE

FIGURE 6: BAS PROGRAMMING EXAMPLE 

©2002 Technical Enterprises, Inc. Reproduction of this document without permission is prohibited.



TECHNICAL SUPPORT • FEBRUARY 2002

caller’s addressing mode. Note: Code
changes from the previous example are
marked in bold.

MVS/ESA THROUGH Z/OS “STUB”

The example in Figure 9 provides support
for the S/370ESA instruction set and still
operates in the z/OS environment. In this
example, use the BAKR/PR instruction
combination to store and recall the GPRs
and the caller’s environment. For recovery
purposes and future planning, I suggest that
you follow the IBM recommendation of
placing the character string of ‘F1SA’ at off-
set +04 in your save area.

Since you are not using the SAVE and
RETURN macros you loose the eye-
catcher that included the program name
and date and time stamp that you had pre-
viously. To provide a similar facility, the
branch at the beginning of the program
furnishes an eye-catcher.

Z/OS “STUB” FOR
REENTRANT PROGRAMS

The code example in Figure 9 provides
support for a z/OS reentrant program stub.
Many systems programs, such as MVS
exits, require reentrancy. A reentrant pro-
gram must not modify its local storage or
face an OS storage violation usually seen
as an MVS ABEND. This program uses
the BAKR/PR instruction combination to
store and recall the GPRs and the caller’s
environment. For recovery purposes and
future planning, I suggest that you follow
the IBM recommendation of placing the
character string of ‘F1SA’ at offset +04 in
your save area.

Since you are not using the SAVE and
RETURN macros you lose the eye-catcher
that included the program name and date
and time stamp that you had previously. To
provide a similar facility, the branch at the
beginning of the program furnishes an eye-
catcher. If this type of program requires
local storage, then the routine must obtain
storage using the GETMAIN or STORAGE
services provided by the operating system.
You should obtain the storage after initial
housekeeping is complete and before any
processing. Once the storage is obtained, it
needs to be cleared since its contents are
unknown. MVS automatically clears stor-
age only when it is obtained on a page
boundary. Figure 9 uses the IBM STOR-

AGE macro to obtain local storage as a
GETMAIN does. Before returning to the
OS, you can issue the STORAGE macro
again with the “RELEASE” parameter to
free it. Note that the STORAGE “OBTAIN”
is conditionally acquired, so R15 must be
checked to ensure the storage requested was
successfully obtained. Also, the use of the
RES= tells the STORAGE macro to obtain
storage wherever the program is resident
(RMODE 24). In this case, it will be below
the line. If your program’s RMODE ever
changes, then the STORAGE location will

automatically change. To clear the locally
obtained storage, the program uses the
MVCL convention. After it is cleared, the
required =CL04’F1SA’ is placed at offset
+04 in the save area mapped by the Storage
DSECT. You will also note the use of the
SYSSTATE macro.

The SYSSTATE macro sets global vari-
ables used by the STORAGE macro and
other system macros that support AR Mode
execution. In other words, if the routine in
Figure 9 ever needs to run in AR Mode, just
change the to “SYSSTATE ASCENV=P” to

PGM1   AMODE 31           Addressing Mode
PGM1   RMODE ANY � Residency Mode
PGM1   CSECT

Using *,R15 
SAVE (14,12),,’PGM1 &SYSDATE. &SYSTIME.’ Add eye catcher
LR   R12,R15
DROP R15
USING PGM1,R12
BSM  R14,0          Save caller’s Addressing mode
ST   R14,12(,R13)    Resave r14               

*  Ensure this program is executing in AMODE(31)
L   R14,AMODE31       Point to 31 bit address  
BSM 0,R14          Set the Addressing mode to 31 Bit
B  PastSet31

AMODE31 DC  0f’0’,AL04(PastSet31+X’80000000’) Amode(31)       
PastSet31    equ  *

ST  R13,SAVEAREA+04  Forward Pointer            
LA   R10,SAVEAREA     Backwards Pointer          
ST  R10,08(R13)      Save area address          
LR  R13,R10         R13 points to our save area 

*   Do something here
RETURN   EQU   *  

L  R13,SAVEAREA+04  Get callers save area
L  R15,EXITRC     Load Return code

*  This section acts like we issued the RETURN macro except
*  we do a ‘BSM’ verses a “BR” in order maintain the callers
*  addressing mode in the return

L   R14,12(0,R13) Restore reg 14
LM R0,R12,20(R13) Restore the registers  
OI 15(R13),X’01’ Set return indication  
BSM 0,R14 Return to caller  

SAVEAREA DS  20F          SAVE AREA                    
EXITRC   DC  F’0’         RETURN CODE                 

LTORG ,           DROP LITERALS HERE
YREGS ,             REGISTER EQUATES
END MYPGM 

� Set RMODE, or Residency Mode, to ANY. This means that you must give careful consideration to any part of your
routine that may require execution in 24-bit mode, like an I/O operation or a DCB. You may want to user “RMODE 24”
as a standard to avoid this situation.

FIGURE 8: MVS/XA ASSEMBLER AMODE(31) PROGRAMMING STUB

PGM1   CSECT
SAVE (14,12)           SAVE CALLERS REGISTERS   
LR  R12,R15          LOAD ADDRESS OF CALLING ROUTINE OUR BASE
USING PGM1,R12           ADDRESS ROUTINE              
ST   R13,SAVEAREA+04       FORWARD POINTER              
LA  R10,SAVEAREA          BACKWARDS POINTER            
ST   R10,08(R13)           SAVE AREA ADDRESS            
LR   R13,R10            R13 POINTS TO OUR SAVE AREA  

*    (DO SOMETHING HERE)
RETURN  EQU  *                                                  

L   R13,SAVEAREA+04      GET CALLERS SAVE AREA     
L    R15,EXITRC           LOAD RETURN CODE
RETURN (14,12),RC=(R15)    RETURN TO OS WITH RETURN CODE

SAVEAREA DS  20F                  SAVE AREA                    
EXITRC  DC   F’0’               RETURN CODE                 

LTORG ,                  DROP LITERALS HERE
YREGS                     REGISTER EQUATES
END  MYPGM

FIGURE 7: MVS ASSEMBLER PROGRAMMING STUB 

“=AR” and the subsequent macros will
expand in their AR mode format. Prepare
for the future now!

When local storage is obtained, the loca-
tion can be specified by the sub-parameter
of “LOC=” on the STORAGE macro. If
“LOC=RES”, then where ever this pro-
gram resides when executed is where the
storage is obtained. The ability to specify
the location of the storage area simplifies
several issues. In particular, if RMODE(24)
is required of any program the local private
storage is obtained in “below-the-line” or
in 24-bit addressable storage. Likewise, if
RMODE(ANY) is specified the local pri-
vate storage is obtained “above-the-line”
(typically).

I wish you luck in preparing to meet the
challenge of programming in the z/OS
environment. The OS enhanced instructions
and improved assembler make coding,
debugging, and maintenance simpler. Part II
will explore additional facilities found in
the HLA and the relatively new HLA
ToolKit — in particular the HLA structured
assembler macros.  

Mark Hedges is a product manager at Landmark
Systems Corporation in Reston, VA. He can be
contacted via email at mhedges@landmark.com
or at (703) 464-1300.

PGM1  AMODE 31               Addressing Mode
PGM1  RMODE ANY                 Residency Mode 
PGM1   CSECT                    Start of routine

SYSSTATE ASCENV=P         Primary mode macro expansions
BAKR R14,0              Save Callers Environment 
MSTA R0                    save R0 and R1 in MSTA
LAE  r12,0000(R15,0)       Base Register
LAE R11,2048(0,R12)       2nd base: R11
LA  R11,2048(,R11)
Using PGM1,R12,R11          Address this routine             
B   Skip_Eyecatcher
DC  C’PGM1  &SYSDATE. &SYSTIME.’,0F’0’

Skip_Eyecatcher equ *
*  Ensure this program is executing in AMODE(31)

EREG R14,r15               Load R14+R15 from the stack
L  R14,AMODE31           Point to 31 bit address  
BSM 0,R14                 Set the mode to 31       
B  PastSet31

AMODE31 DC  0f’0’,AL04(PastSet31+X’80000000’)  Amode(31)       
PastSet31  equ   *

LA  R0,Storage_Size       Size of local storage: an equate
STORAGE OBTAIN,             Obtain Storage                 ++

LENGTH=(R0),          length                         ++
COND=YES,             Conditional                    ++
SP=0,                Subpool                        ++
LOC=RES               Where to allocate

LTR R15,R15                 How did we do?
BNZ Exit                    .Not so well, exit with RC in R15
LR  R13,R1                  Save storage address in R13   
LR  R14,R1                  Clear storage                
LA  R15,Storage_size        Length                       
SLR R1,R1                   Clear r1                     
MVCL R14,R0                  Clear obtained storage area
USING Storage,R13              Map storage area
MVC  04(04,R13),=CL04’F1SA’  As required by OS
XC   EXITC,EXITRC            Clear Return code

*  >>   BAS  R14,Process_Parm        Branch and do something
Return  EQU *                         Finished 

L   R2,EXITRC               save RC since we loose addressing 
LA  R0,Storage_Size         Storage Size 
LR  R1,R13                  Storage address
STORAGE RELEASE,              Free Local Storage            

LENGTH=(R0),                                          
ADDR=(R1)                                             

LR  R15,R2                  Return code saved before      
EXIT  EQU  *

La   R0,3                    Load ESTA code of 3
Esta R0,r0                   Extract the stored versions of R0 and R1
PR    ,                     GO BACK to OS                 

*  Literal Pool
Ltorg  ,

*  Local storage mapping DSECT Storage
Storage        DSECT
SAVEAREA        DS 20F SAVE AREA
EXITRC        DS F RETURN CODE

DS 0D Buffer
DS 4D A little extra room

Storage_size      equ    *-Storage
YREGS  ,                REGISTER EQUATES
CVT DSECT=YES,PREFIX=YES,LIST=NO
END PGM1

FIGURE 9: Z/OS “STUB” FOR REENTRANT PROGRAM 


